```
$EOLCOM //
$TITLE Optimizing Water Resource Use in the TE Basin - Iraq
$OFFSYMXREF OFFSYMLIST OffLISTING OFFUPPER

OPTION LIMROW=000, LIMCOL = 0;

$ONTEXT

* ---------------------------------------------------------------------------------------

  Frank A. Ward:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, NM USA
  e-mail: fward@nmsu.edu

  Dina A. Salman:
  Dept of Agr Economics/Agr Business
  New Mexico State University, Las Cruces, New Mexico USA
  email: dinasalm@nmsu.edu

  Saud A.Amer:
  US Geological Survey - International Water Resources Division
  Reston VA  USA
  email:  samer@usgs.gov

  May 13, 2013

* ---------------------------------------------------------------------------------------
*  Model 3 Simple Farm Management
*
*  Adds a SET for water supply scenario (2 scenarios: normal-dry)
*  Adds data on observed land in production (Thanks to Saud Amer)
*  Normal water use calculated parameter: observed land in prodn multiplied by crop water use
* ---------------------------------------------------------------------------------------

$OFFTEXT
```

```
*************************************   Section 1 *********************************
*                                        Sets                                       *
*****************************************************************************************

SETS

i    province

/
  1-Mousil
  2-Basra
/

k    crop

/  1-wheat
   2-cotton
/

s   hydrologic water supply scenario

/
normal
dry

/

*********************************  Section 2 ************************************
*                                      Data                                       *
*****************************************************************************************

table  Bc(i, k)    crop water demand (ET) (1000s cubic meters per Ha = 10ths meters depth)

                1-wheat          2-cotton
 1-mousil         11.9             18.0
 2-Basra          13.5             21.4

*Bc(k) =  0.5 * Bc(k)

table    Yield_p(i,k)   Crop Yield tons per HA (proportional to ET)

               1-wheat         2-Cotton
```

```
1-Mousil        1.39            2.32
2-Basra         1.57            2.75


* economic data


parameter   Price_p(k)        Crop Prices ($ US per ton)

/1-Wheat        225
 2-Cotton       906
/

table Cost_p(i,k)     Production Costs excluding water ($US per Ha)

               1-wheat   2-cotton
1-Mousil         200      1286
2-Basra          207      1345

;

Table land_p(i,k)   observed land in prodn from year 2006 (1000 Ha)

                1-wheat        2-cotton
1-Mousil          47.4           0.45
2-Basra           16.9           0.00
;

Parameter Net_revenue_p(i,k)  calculated net revenue per unit land ($ US per Ha)
;

Net_revenue_p(i,k) = Price_p(k) * Yield_p(i,k) - Cost_p(i,k);


parameter Wat_supply_p(s)   total water used: full supply = year 2006 (million cubic meters per year)
;
Wat_supply_p('normal') = sum((i,k), land_p(i,k) * Bc(i,k));   // calculates observed water use
Wat_supply_p('dry')    = 0.5 * wat_supply_p('normal');        // dry: half of observed water use

*****************************  SECTION 3 ****************************************
*                              Variables                                          *
*********************************************************************************
```

```
positive variables

hectares_v     (i,k,s)   land in production by crop-province-scen    (1000 Ha - marginal is $US per Ha)
T_hectares_v   (i,  s)   total land in prodn by province-scen       (1000 Ha - marginal is $US per Ha)
uses_crop_v    (i,k,s)   total water use by crop-province-scen      (million m^3 - marginal is $US per 1000 m^3)
uses_v         (    s)   total water use by scen                    (million m^3 - marginal is $US per 1000 m^3)

variables

ag_ben_k_v     (i,k,s)   total farm income by crop-province-scen    ($US 1000s - no marginals shown)
ag_ben_v       (    s)   total farm income by scen                  ($US 1000s - no marginals shown)

ag_ben_normal_v          total farm income (objective - full water)    ($US 1000s - no marginals shown)
ag_ben_dry_v             total farm income (objective - reduced water) ($US 1000s - no marginals shown)


*****************************   Section 4  ****************************************
*                               Equations                                          *
*********************************************************************************

Equations

//equations declared

T_hectares_e   (i,  s)    total land in production by province-scen  (1000 Ha - marginal is $US per Ha)
Uses_crop_e    (i,k,s)    total water use by crop-province-scean     (million m^3 - marginal is $US per 1000 m^3)
uses_e         (    s)    total water use by scen                    (million m^3 - marginal is $US per 1000 m^3)

ag_ben_k_e     (i,k,s)    total farm income by crop-province-scen    ($US 1000s - no marginals shown)
ag_ben_e       (    s)    total farm income by scen                  ($US 1000s - no marginals shown)

ag_ben_normal_e          total farm income (obj 1-full water)       ($US 1000s - no marginals shown)
ag_ben_dry_e             total farm income (obj 2-reduced water)    ($US 1000s - no marginals shown)
;

// equations defined (using above names with algebraic formulas)

T_hectares_e   (i,  s)..  T_hectares_v(i,  s)   =e=  sum(k,   hectares_v   (i,k,s));   // land in prodn by prov-scen
Uses_crop_e    (i,k,s)..  uses_crop_v (i,k,s)   =e=  Bc(i,k) *  hectares_v (i,k,s);    // water use by crop-prov-scen
uses_e         (    s)..  uses_v      (    s)   =e=  sum((i,k),uses_crop_v (i,k,s));       // total water use by scen
ag_ben_k_e     (i,k,s)..  ag_ben_k_v  (i,k,s)   =e=  net_revenue_p(i,k) * hectares_v(i,k,s); // income crop-prov-scen
```

```
ag_ben_e       (    s)..  ag_ben_v    (    s)   =e=  sum((i,k),  ag_ben_k_v(i,k,s));  // total farm income by scen
ag_ben_normal_e       ..  ag_ben_normal_v       =e=  ag_ben_v('normal');      // obj fn 1: income full water scen
ag_ben_dry_e          ..  ag_ben_dry_v          =e=  ag_ben_v(   'dry');      // obj fn 2: income reduced water scen

*************** Section 5 ************************************************************************
*  Labels and defines all models used                                                            *
*  Each model has one objective function                                                                 *
*****************************************************************************************************

model TE_03 /all/ ;

*********************************  Section 6   ***************************************
*                                 BOUNDS                                              *
*  Bounding shows a positive shadow price for each limiting resource                  *
*****************************************************************************************

uses_v.up    (    s) = wat_supply_p(s);  // upper bound on total water available (2 scenarios)
hectares_v.up(i,k,s) = land_p(i,k);      // can allocate no more land to crops than full water (2006)

*hectares_v.lo('1-mousil','1-wheat') =  10; // food security requirement
*uses_crop_v.lo('wheat')  = 100;            // lower bound on wter assigned to wheat

*************** Section 7 ************************************************************************
*               SOLVE THE MODEL                                                               *
*****************************************************************************************************

* model 1
solve TE_03 using nlp maximizing ag_ben_normal_v;

* model 2
uses_crop_v.fx(i,k,'normal') = uses_crop_v.l(i,k,'normal');  // save optimized water use from 1st model solution

solve TE_03 using nlp maximizing ag_ben_dry_v;

// THE END
```